

# A CO<sub>2</sub> monitor as an introductory microelectronics project helping to slow-down the spread of the corona virus and ensuring a healthy learning and working environment

A. Köhn-Seemann, [alf.koehn@gmail.com](mailto:alf.koehn@gmail.com)

2020-09-21

## Abstract

This paper describes the setup of a simple yet reliable CO<sub>2</sub> monitor which is based on open-source microelectronics hardware. The monitor is intended to be used in class rooms, lecture halls or offices and can be constructed as a joint students project. It was motivated by recent discussions on the role of aerosols being part of exhaled air to spread the corona virus. The aerosol concentration in air is correlated with the CO<sub>2</sub> concentration. Measuring the latter can thus help to slow-down the spread of the corona-virus. The program code used for the CO<sub>2</sub> monitor and this documentation is available as a GitHub repository to allow to updates and improvements.

License for this document: CC BY-SA 4.0

## 1 Introduction

It is generally accepted that the CO<sub>2</sub> concentration in a class room has an influence on students' activities, their ability to study and learn [1, 2], or on their health and thus attendance [3]. The same applies of course to office environments [4]. The major source of CO<sub>2</sub> in a class room is the exhaled air of the students (and teachers). It thus increases over time but can also be relatively easy controlled by proper ventilation. Monitoring the CO<sub>2</sub> concentration over time provides thus a simple way to ensure an productive and healthy learning environment.

In addition to  $\text{CO}_2$ , exhaled air consists of aerosols (among other things). In preliminary studies, it has been recently discovered that the aerosols of patients being infected with Sars-CoV-2, might contain viable virus concentrations which are large enough to cause further infections if somebody else inhales those aerosols [5–7]. With half-life periods of the virus on aerosols on the order of 1 hour [8], it becomes evident that proper ventilation, strongly reducing the aerosol concentration, can help to prevent hidden infections, i.e. infections where the infected person is not (yet) aware of their infection but already contagious. Since aerosols and  $\text{CO}_2$  are both parts of exhaled air, measuring the  $\text{CO}_2$  concentration in a room provides an easy accessible approximation for the aerosol concentration [9].

Here we present a simple and cost effective, yet reliable way to monitor the  $\text{CO}_2$  concentration. Widely available microelectronic components are used which can be easily programmed via open source software platforms allowing to modify and extend the example presented in this paper. Students can build the detectors in class as a joint project which might serve to raise interest in electronics or the underlying physical and chemical processes [10].

This work was inspired by a project of the *Hochschule Trier* [11], where the design and construction of a  $\text{CO}_2$  measuring device is suggested as a students' project, allowing to discuss a variety of scientific topics during the course of the project. In addition, a few posts from different forums served as an inspiration [12–15]. Furthermore, a small number of GitHub repositories using the same  $\text{CO}_2$  are available [16–18] (we would like to recommend the interested reader in particular to the repository by paulvha [18] as it contains a rather large number of examples).

## 2 The $\text{CO}_2$ monitor

The  $\text{CO}_2$  monitor is based on the microelectronic sensor SCD30 which measures the  $\text{CO}_2$  concentration and also provides measurements of the ambient temperature and relative humidity [19]. Using Arduino as a programming language and some microcontroller, it is straightforward to get the sensor running and outputting data, thanks to the examples available in the libraries provided by SparkFun [20]. Using the Arduino IDE [21], which is available for all major operating systems, the corresponding libraries can be simply included via the library manager.

To make the  $\text{CO}_2$  monitor visually appealing, we decided to output the measurement to an OLED display (which is also very cheap and available in a large variety of sizes and configurations). Due to the widespread usage of such displays, they can also be directly included via the library manager

in the Arduino IDE. In addition to just showing some numbers, we have included a red LED which lights up as soon as some threshold value of the CO<sub>2</sub> concentration is reached, indicating the need for ventilation. One could also think of a traffic light design, where first a yellow LED lights up at a slightly lower threshold value. The *Federation of European Heating, Ventilation and Air Conditioning associations (REHVA)* recommend to issue a warning, corresponding to an orange light, when a value of 800 ppm is reached and prompt to trigger some action like ventilation, corresponding to a red light, when 1000 ppm are reached [22]. The *Federal Ministry of Labour and Social Affairs* of Germany also states a threshold value of 1000 ppm that should not be passed [23]. Note that a value of 400 ppm is the typical CO<sub>2</sub> concentration of air.

As controller we decided to use the low-cost open source NodeMCU board [24], as it offers enough flexibility to further extend the functionality of the CO<sub>2</sub> monitor. Of particular interest might be the WiFi capability allowing for example to write the measured values to a web-server where they can then accessed via a web-browser or an app on a smartphone.

A prototype of the CO<sub>2</sub> monitor is shown in Fig. 1. As one can see, it is not enclosed in some box to still allow easy access for modifications. The idea of this prototype was rather to show that the general principle of the CO<sub>2</sub> monitor is working and not to provide a polished final product. The prototype is ready to be used in a class room or lecture hall, although it might be worth to mount everything into a box which is not only visually more appealing but provides also some protection.

### 3 Required parts

The CO<sub>2</sub> monitor as presented here consists of a number of parts for which it is not important to use the exact same model. The only component which should not be replaced is the CO<sub>2</sub> measuring device, the SCD30. Note that the program code discussed in Sec. 6 is tailored for the NodeMCU ESP8266, replacing that component would thus require small adjustments to the code.

The parts used for the prototype of the CO<sub>2</sub> monitor are listed in Table 3. The display can be easily replaced by an OLED of larger size. One could also use multiple displays, which would require to take care of proper addressing the display and thus add a little bit of complexity to the code (and the assembly).

The usage of a breadboard was motivated by educational purposes as this allows very easy assembly without the need to solder anything. It can, however, directly be replaced by a stripboard or completely omitted and use




Figure 1: Assembled and working prototype of the CO<sub>2</sub> monitor, (*left*) with a measured CO<sub>2</sub> concentration below the threshold and (*right*) above it (note the red LED).

only cables or pin headers (which would require some soldering).

Note that the prices as listed in the table can be pushed down (significantly for some of the components) when ordering larger quantities.

For the prototype design of the CO<sub>2</sub> monitor we have decided to leave out a proper casing. One could either use a standard-sized case, or design one and print it for example on a 3D printer or re-use/recycle some old boxes. It is however important to correctly position the SCD30 inside the box: as described in a manufacturer’s document [25], the sensor is ideally placed as close as possible to the box’s outer shell and to a large opening to be properly exposed to the ambient. The box should be as small as possible to get fast response times to changes in the ambient air. The SCD30 should also be isolated from direct air flow, as the corresponding changes in pressure (due to the air flow) would lead to increased noise and thus reduced accuracy in the measurements. It is also recommended to not directly place the sensor above heat sources like for example microcontrollers.

## 4 The CO<sub>2</sub> sensor

The SCD30 has been chosen because it performs direct measurements of the CO<sub>2</sub> concentration. Cheaper sensors often measure the concentration of volatile organic compounds (VOC) and then assume a correlation between the two quantities. This can, however, lead to wrong values of the

Element	Quantity	Price
SCD30 (CO <sub>2</sub> sensor)	1	60 €
NodeMCU EPS8266	1	8 €
0.91" OLED display	1	5 €
red LED	1	0.2 €
220 Omega resistor	1	.1 €
mini breadboard	1	4 €
breadboard cables	10	4 €
pin header	1	0.5 €
micro USB cable	1	3 €

Table 1: Components used for the CO<sub>2</sub> monitor as presented in this paper (note that the prices were obtained in 09/2020 and may vary).

CO<sub>2</sub> concentration since VOC can be emitted from a variety of chemicals. Although VOCs are also known to cause health problems, here we are explicitly interested in the CO<sub>2</sub> concentration, as discussed in Sec. 1. For a discussion about monitoring VOC and CO<sub>2</sub> concentration with self-assembled devices we would like to point the interested reader to e.g. Ref. [26].

## 4.1 Technical specifications

According to the datasheet of the SCD30 [19], the CO<sub>2</sub> sensor has a measurement range of 0 – 40,000 ppm with an accuracy of  $\pm 30$  ppm. The supply voltage needs to be between 3.3 and 5 V which allows to use a variety of microcontrollers. The drawn current is specified to be on average 19 mA with a maximum value of 75 mA. With a sensor lifetime of 15 years, the SCD30 offers a reliable system to permanently monitor the CO<sub>2</sub> concentration.

## 4.2 Nondispersive infrared technique

The CO<sub>2</sub> concentration is measured using the so-called *nondispersive infrared* technique (NDIR). It is the most common sensor type used in industry to measure the CO<sub>2</sub> concentration. Its principle is sketched in Fig. 2. A light source emits infrared light which travels through a tube filled with a sample of the surrounding air. The spectrum of the emitted light includes the 4.26  $\mu\text{m}$  absorption band of CO<sub>2</sub> which is unique to the typical components of air and the light is absorbed by them. At the end of the tube, the remaining light hits an optical filter that allows only that specific wavelength of 4.26  $\mu\text{m}$  to pass. A detector then collects the remaining light. The difference between




Figure 2: Sketch of a sensor using the nondispersive infrared technique to measure CO<sub>2</sub> concentration.

the amount of light emitted by the source and received by the detector is due to the CO<sub>2</sub> molecules in the tube which then allows to calculate the CO<sub>2</sub> concentration.

Using folded optics, i.e. waveguides, for the tube and diodes for the infrared source and detector, allows to small sizes of the overall sensor on the order of just a few centimeters.

## 5 Assembly




Figure 3: Schematic of a prototype of the CO<sub>2</sub> monitor.

The CO<sub>2</sub> monitor can be assembled in various ways, here we will restrict ourselves to the case of a simple prototype design on a breadboard as shown in Fig. 3. The connection between the NodeMCU (with the ESP8266) and

the SCD30 sensor is as follows:

NodeMCU		SCD30
GND	→	GND
3.3 V	→	VIN
D2/GPI04	→	RX/SDA
D1/GPI05	→	TX/SCL
GND	→	SEL

The NodeMCU then needs to be connected to the OLED display as follows:

NodeMCU		OLED display
GND	→	GND
3.3 V	→	VCC
D2/GPI04	→	SDA
D1/GPI05	→	SCL

It is of course also possible to directly connect the respective **SDA** and **SCL** pins of the OLED and the SCD30, as shown in Fig. 3, instead of connecting those pins between the SCD30 and the NodeMCU. The red LED is connected with its anode, the longer leg, to pin **D8/GPI015** of the NodeMCU and with its cathode, the shorter leg, via a 220 Ohm resistor (to limit the current) to ground.

## 6 The program code

Arduino is used as programming language in this project due to its widespread usage and large numbers of libraries available for various hardware components. The Arduino IDE library manager allows to directly install a proper Arduino library for the SCD30, alternatively the library is available as a GitHub repository [20]. For a tutorial on how to install libraries within the Arduino IDE, see Ref. [27]. As for the NodeMCU and the OLED display, the Arduino IDE library manager is able to provide the required libraries.

The source code for the CO<sub>2</sub> monitor as described in this paper is available on GitHub [28], in order to be able to update and extend it. Nevertheless, we have also included the code in this paper, to provide a complete description of the project. The include statements of the code are listed in Listing 1. The `Adafruit_GFX.h` and `Adafruit_SSD1306.h` libraries are used for the

OLED display and are required to be installed via the library manager of the Arduino IDE beforehand (alternatively, they are also available on GitHub [29] for manual installation). Note that the display size in pixels needs to be set correctly and can vary. The `SparkFun_SCD30_Arduino_Library.h` also needs to be installed via the library manager (or manually from the GitHub repository [20]).

```

1 #include <ESP8266WiFi.h>
2 #include <Hash.h> // for SHA1 algorithm (for Font Awesome)
3 #include <ESPAsyncTCP.h>
4 #include <ESPAsyncWebServer.h>
5 #include <Wire.h> // for I2C communication
6 #include <Adafruit_GFX.h> // for writing to display
7 #include <Adafruit_SSD1306.h> // for writing to display
8 #include "SparkFun_SCD30_Arduino_Library.h"
9
10 #include "Webpageindex.h" // webpage content, same folder as .ino
11     file
12 // Replace with your network credentials
13 const char* ssid = "WIFI_SSID";
14 const char* password = "WIFI_PASSWORD";
15
16 // activate debugging
17 // true: print info + data to serial monitor
18 // false: serial monitor is not used
19 #define DEBUG true
20
21 #define CO2_CRITICAL 1500 // threshold for warning
22 #define WARNING_DIODE_PIN D8 // NodeMCU pin for red LED
23
24 #define MEASURE_INTERVAL 10 // seconds, minimum: 2
25
26 #define SCREEN_WIDTH 128 // OLED display width in pixels
27 #define SCREEN_HEIGHT 32 // OLED display height in pixels
28
29 // OLDE reset pin, 4 is default (-1 if sharing Arduino reset pin)
30 // using NodeMCU, we have to use LED_BUILTIN
31 #define OLED_RESET LED_BUILTIN
32 // Declaration for an SSD1306 display connected to I2C (SDA, SCL pins)
33 Adafruit_SSD1306 display(SCREEN_WIDTH, SCREEN_HEIGHT, &Wire, OLED_RESET);
34
35 SCD30 airSensor;
36
37 // current temperature & humidity, updated in loop()
38 float temperature_web = 0.0;
39 float humidity_web = 0.0;
40 float co2_web = 0.0;
41
42 // Create AsyncWebServer object on port 80
43 AsyncWebServer server(80);
44
45 // use "unsigned long" for variables that hold time
46 // --> value will quickly become too large for an int
47 unsigned long previousMilliseconds = 0; // will store last time scd30 was
    updated
48
49 // update scd30 readings every MEASURE_INTERVAL seconds
50 const long interval = MEASURE_INTERVAL*1000;

```



```

51
52 // read html into string
53 String webpage = index_html;
54
55 // Replaces placeholder with DHT values
56 String processor(const String& var){
57   //Serial.println(var);
58   if(var == "CO2"){
59     return String(co2_web);
60   }
61   else if(var == "TEMPERATURE"){
62     return String(temperature_web);
63   }
64   else if(var == "HUMIDITY"){
65     return String(humidity_web);
66   }
67   return String();
68 }

```

Listing 1: Loading the required libraries.

Listing 2 shows the setup of the code, where the serial monitor is initialized, followed by the diode, the OLED display, and finally the SCD30.

```

1 void setup(){
2   if (DEBUG == true) {
3     // initialize serial monitor at baud rate of 115200
4     Serial.begin(115200);
5     Serial.println("Using SCD30 to get: CO2 concentration, temperature,
6     humidity");
7   }
8   // initialize I2C
9   Wire.begin();
10
11   // initialize LED pin as an output
12   pinMode(WARNING_DIODE_PIN, OUTPUT);
13
14   // Connect to Wi-Fi
15   WiFi.begin(ssid, password);
16   if (DEBUG == true)
17     Serial.println("Connecting to WiFi");
18   while (WiFi.status() != WL_CONNECTED) {
19     delay(1000);
20     if (DEBUG == true)
21       Serial.print(".");
22   }
23   if (DEBUG == true)
24     Serial.println(WiFi.localIP());
25
26   // SSD1306_SWITCHCAPVCC: generate display voltage from 3.3V internally
27   if(!display.begin(SSD1306_SWITCHCAPVCC, 0x3C)) { // Address 0x3C for 128
28     //x32
29     if (DEBUG == true)
30       Serial.println(F("SSD1306 allocation failed"));
31     for(;;); // Don't proceed, loop forever
32   }
33
34   // Show initial display buffer contents on the screen --
35   // the library initializes this with an Adafruit splash screen.
36   display.display(); // initialize display

```

```

36 // library will show Adafruit logo
37 delay(2000); // pause for 2 seconds
38 display.clearDisplay(); // clear the buffer
39 display.setTextSize(1); // has to be set initially
40 display.setTextColor(WHITE); // has to be set initially
41
42 // move cursor to position and print text there
43 display.setCursor(2,5);
44 display.println("CO2 monitor");
45 display.println("twitter.com/formbar");
46
47 display.display(); // write display buffer to display
48
49 // turn warning LED on and off to test it
50 digitalWrite(WARNING_DIODE_PIN, HIGH);
51 delay(2000);
52 digitalWrite(WARNING_DIODE_PIN, LOW);
53
54 // initialize SCD30
55 // SCD30 has data ready every two seconds
56 if (airSensor.begin() == false) {
57     if (DEBUG == true)
58         Serial.println("Air sensor not detected. Please check wiring. Freezing
59         ...");
60     while (1)
61         ;
62 }
63
64 // Route for root / web page
65 server.on("/", HTTP_GET, [](AsyncWebServerRequest *request){
66     request->send_P(200, "text/html", index_html, processor);
67 });
68 server.on("/co2", HTTP_GET, [](AsyncWebServerRequest *request){
69     request->send_P(200, "text/plain", String(co2_web).c_str());
70 });
71 server.on("/temperature", HTTP_GET, [](AsyncWebServerRequest *request){
72     request->send_P(200, "text/plain", String(temperature_web).c_str());
73 });
74 server.on("/humidity", HTTP_GET, [](AsyncWebServerRequest *request){
75     request->send_P(200, "text/plain", String(humidity_web).c_str());
76 });
77
78 // Start server
79 server.begin();

```

Listing 2: Setup code.

The main code is given in Listing 3. First, the data is obtained from the SCD30 sensor and then passed to a function outputting it to the serial monitor and then to another function, printing it on the OLED display. Listings 4 and 5 show the code for the two latter functions.

```

1 void loop(){
2
3     float
4     co2_new,
5     temperature_new,
6     humidity_new;
7

```

```

8   unsigned long currentMilliseconds;
9
10  // get milliseconds passed since program started to run
11  currentMilliseconds = millis();
12
13  if (currentMilliseconds - previousMilliseconds >= interval) {
14      // save the last time you updated the DHT values
15      previousMilliseconds = currentMilliseconds;
16
17      if (airSensor.dataAvailable()) {
18          // get data from SCD30 sensor
19          co2_new      = airSensor.getCO2();
20          temperature_new = airSensor.getTemperature();
21          humidity_new  = airSensor.getHumidity();
22
23          // print data to serial console
24          if (DEBUG == true)
25              printToSerial(co2_new, temperature_new, humidity_new);
26
27          // print data to OLED display
28          printToOLED(co2_new, temperature_new, humidity_new);
29
30          co2_web      = co2_new;
31          temperature_web = temperature_new;
32          humidity_web  = humidity_new;
33      }
34
35      // if CO2-value is too high, issue a warning
36      if (co2_web >= CO2_CRITICAL) {
37          digitalWrite(WARNING_DIODE_PIN, HIGH);
38      } else {
39          digitalWrite(WARNING_DIODE_PIN, LOW);
40      }
41  }
42 }

```

Listing 3: Main loop which is executed repeatedly.

```

1 void printToSerial( float co2, float temperature, float humidity) {
2     Serial.print("co2(ppm):");
3     Serial.print(co2, 1);
4     Serial.print(" temp(C):");
5     Serial.print(temperature, 1);
6     Serial.print(" humidity(%):");
7     Serial.print(humidity, 1);
8     Serial.println();
9 }

```

Listing 4: Function which prints data to the serial console.

```

1 void printToOLED( float co2, float temperature, float humidity) {
2     int
3     x0, x1;          // to align output on OLED display vertically
4
5     x0  = 2;
6     x1  = 86;
7
8     display.clearDisplay();
9     display.setCursor(x0,5);
10    display.print("CO2 (ppm):");

```

```

11 display.setCursor(x1,5);
12 // for floats, 2nd parameter in display.print sets number of decimals
13 display.print(co2, 0);
14
15 display.setCursor(x0,15);
16 display.print("temp. ( C)");
17 display.setCursor(x0+7*6,15);
18 display.write(167);
19 display.setCursor(x1,15);
20 display.print(temperature, 1);
21
22 display.setCursor(x0,25);
23 display.print("humidity (%):");
24 display.setCursor(x1,25);
25 display.print(humidity, 1);
26
27 display.display();
28 }

```

Listing 5: Function which prints data to the OLED display.

## Acknowledgments

The author is indebted to the efforts of the open-source software community.

## References

- [1] D. Twardella, W. Matzen, T. Lahrz, R. Burghardt, H. Spegel, L. Hendrowarsito, A. C. Frenzel, and H. Fromme. Effect of classroom air quality on students' concentration: results of a cluster-randomized cross-over experimental study. *Indoor Air*, 22:378, 2012, doi:10.1111/j.1600-0668.2012.00774.x.
- [2] S. Gaihre, S. Semple, J. Miller, S. Fielding, and S. Turner. Classroom Carbon Dioxide Concentration, School Attendance, and Educational Attainment. *J. Sch. Health*, 85:569–574, 2014, doi:10.1111/josh.12183.
- [3] D. G. Shendell, R. Prill, W. J. Fisk, M. G. Apte, D. Blake, and D. Faulkner. Associations between classroom co<sub>2</sub> concentrations and student attendance in washington and idaho. *Indoor Air*, 14:333, 2004, doi:10.1111/j.1600-0668.2004.00251.x.
- [4] J. G. Allen, P. MacNaughton, U. Satish, S. Santanam, J. Vallarino, and J. D. Spengler. Associations of Cognitive Function Scores with Carbon Dioxide, Ventilation, and Volatile Organic Compound Exposures in Office Workers: A Controlled Exposure Study of Green and Conventional

- Office Environments. *Environmental Health Perspectives*, 6:805, 2016, doi:10.1289/ehp.1510037.
- [5] J. A. Lednický, M. Lauzardo, Z. Hugh Fan, A. S. Jutla, T. B. Tilly, M. Gangwar, M. Usmani, S. N. Shankar, K. Mohamed, A. Eiguren-Fernandez, C. J. Stephenson, Alam. Md. M., M. A. Elbadry, J. C. Loeb, K. Subramaniam, T. B. Waltzek, K. Cherabuddi, J. G. Morris Jr., and C.-Y. Wu. Viable sars-cov-2 in the air of a hospital room with covid-19 patients. *medRxiv*, 2020, doi:10.1101/2020.08.03.20167395.
  - [6] L. Morawska and D. K. Milton. It is time to address airborne transmission of covid-19. *Clinical Infectious Diseases*, 2020, doi:10.1093/cid/ciaa939.
  - [7] M. A. Kohanski, L. J. Lo, and M. S. Waring. Review of indoor aerosol generation, transport, and control in the context of covid19. *International Forum of Allergy & Rhinology*, 2020, doi:10.1002/alr.22661.
  - [8] N. v. Doremalen, T. Bushmaker, D. H. Morris, M. G. Holbrook, A. Gamble, B. N. Williamson, A. Tamin, J. L. Harcourt, N. J. Thornburg, S. I. Gerber, J. O. Lloyd-Smith, E. de Witt, and V. J. Munster. Aerosol and Surface Stability of SARS-CoV-2 as Compared with SARS-CoV-1. *N. Engl. J. Med.*, 382:1564–1567, 2020, doi:10.1056/NEJMc2004973.
  - [9] A. Hartmann and M. Kriegel. Risk assessment of aerosols loaded with virus based on  $\text{CO}_2$ -concentration. *Preprint server of the Technische Universität Berlin*, 2020, doi:10.14279/depositonce-10362.
  - [10] B. Watzka and R. Girwidz. Ist die Luft zu schlecht zum Lernen? Nicht-dispersive IR- $\text{CO}_2$ -Gassensoren im Physikunterricht. *Physik und Didaktik in Schule und Hochschule*, 10:22–33, 2011.
  - [11] Hochschule Trier Umwelt-Campus Birkenfeld. Website: Covid-19 Prävention:  $\text{CO}_2$ -Messung und bedarfsorientierte Lüftung (accessed: 2020-09-13). <https://www.umwelt-campus.de/forschung/projekte/iot-werkstatt/ideen-zur-corona-krise>.
  - [12] hackster.io (user Brian Boyles). Website: Homemade  $\text{CO}_2$  Sensor Unit (accessed: 2020-09-13). <https://www.hackster.io/bfboyles/homemade-co2-sensor-unit-22a9d8>.
  - [13] Arduino Forum (user metropol). Website: Arduino forum:  $\text{CO}_2$  level classroom sensor (accessed: 2020-09-13). <https://forum.arduino.cc/index.php?topic=702131>.

- [14] Home Assistant Forum (user Ombra). Website: Co2 monitoring with scd30 sensor and ha integration (accessed: 2020-09-13). <https://community.home-assistant.io/t/co2-monitoring-with-scd30-sensor-and-ha-integration/216708>.
- [15] Raspberry Pi Forum (user Zentris). Website: In Entwicklung: Co2-Sensor (SCD30) mit ESP8266 und Display SSD1306 (kurze Vorstellung) (accessed: 2020-09-13). <https://forum.raspberrypi.de/forum/thread/44717-in-entwicklung-co2-sensor-scd30-mit-esp8266-und-display-ssd1306-kurze>
- [16] Kaspar Metz. Coro<sub>2</sub>sens. <https://github.com/kmetz/coro2sens>, 2020.
- [17] Netzbasteln. Co<sub>2</sub>narienvogel. <https://github.com/netzbasteln/co2narienvogel>, 2020.
- [18] paulvha. paulvha scd30 library. <https://github.com/paulvha/scd30>, 2020.
- [19] Sensirion: The Sensor Company. Datasheet sensirion scd30 sensor module (version 0.94). [https://www.sensirion.com/fileadmin/user\\_upload/customers/sensirion/Dokumente/9.5\\_CO2/Sensirion\\_CO2\\_Sensors\\_SCD30\\_Datasheet.pdf](https://www.sensirion.com/fileadmin/user_upload/customers/sensirion/Dokumente/9.5_CO2/Sensirion_CO2_Sensors_SCD30_Datasheet.pdf).
- [20] SparkFun Electronics. SparkFun SCD30 CO2 Sensor Library. [https://github.com/sparkfun/SparkFun\\_SCD30\\_Arduino\\_Library](https://github.com/sparkfun/SparkFun_SCD30_Arduino_Library), 2020.
- [21] The Arduino team. Software: The open-source arduino ide. <https://www.arduino.cc/en/main/software>.
- [22] Ventilation The Federation of European Heating and Air Conditioning associations (REHVA). Website: Rehva covid-19 guidance document (version 2020-08-03). <https://www.rehva.eu/activities/covid-19-guidance/rehva-covid-19-guidance>.
- [23] Federal Ministry of Labour and Social Affairs Germany. Website: SARS-CoV-2-Arbeitsschutzregel (version 2020-08-20). <https://www.bmas.de/DE/Presse/Meldungen/2020/neue-sars-cov-2-arbeitsschutzregel.html>.
- [24] NodeMCU. Website: Nodemcu documentation (accessed: 2020-09-18). <https://nodemcu.readthedocs.io/en/master/>.

- [25] Sensirion: The Sensor Company. Design-in guidelines for scd30 (version 0.2). [https://www.sensirion.com/fileadmin/user\\_upload/customers/sensirion/Dokumente/9.5\\_CO2/Sensirion\\_CO2\\_Sensors\\_SCD30\\_Assembly\\_Guideline.pdf](https://www.sensirion.com/fileadmin/user_upload/customers/sensirion/Dokumente/9.5_CO2/Sensirion_CO2_Sensors_SCD30_Assembly_Guideline.pdf).
- [26] G. Chiesa, S. Cesari, M. Garcia, M. Issa, and S. Li. Multisensor IoT Platform for Optimising IAQ Levels in Buildings through a Smart Ventilation System. *Sustainability*, 1:5777, 2019, doi:10.3390/su11205777.
- [27] SparkFun Electronic. Website: Installing an Arduino Library. <https://learn.sparkfun.com/tutorials/installing-an-arduino-library>.
- [28] Alf Köhn-Seemann. Co<sub>2</sub> monitor. [https://github.com/alfkoehn/CO2\\_monitor](https://github.com/alfkoehn/CO2_monitor), 2020.
- [29] Adafruit Industries. Adafruit\_ssd1306. [https://github.com/adafruit/Adafruit\\_SSD1306](https://github.com/adafruit/Adafruit_SSD1306), 2020.