A CO<sub>2</sub> monitor as an introductory microelectronics project helping to slow-down the spread of the corona virus and ensuring a healthy learning and working environment

> A. Köhn-Seemann, alf.koehn@gmail.com 2020-10-22

#### Abstract

This paper describes the setup of a simple yet reliable  $\mathrm{CO}_2$  monitor which is based on open-source microelectronics hardware. The monitor is intended to be used in class rooms, lecture halls or offices and can be constructed as a joint students project. It was motivated by recent discussions on the role of aerosols being part of exhaled air to spread the corona virus. The aerosol concentration in air is correlated with the  $\mathrm{CO}_2$  concentration. Measuring the latter can thus help to slow-down the spread of the corona-virus. The program code used for the  $\mathrm{CO}_2$  monitor and this documentation is available as a GitHub repository to allow to updates and improvements.

First full version released on 2020-09-22.

License for this document: CC BY-SA 4.0

### 1 Introduction

It is generally accepted that the  $CO_2$  concentration in a class room has an influence on students' activities, their ability to study and learn [1, 2], or on their health and thus attendance [3]. The same applies of course to office environments [4]. The major source of  $CO_2$  in a class room is the exhaled air of the students (and teachers). It thus increases over time but can also be relatively easy controlled by proper ventilation. Monitoring

the CO<sub>2</sub> concentration over time provides thus a simple way to ensure a productive and healthy learning environment.

In addition to  $CO_2$ , exhaled air consists of aerosols (among other things). In preliminary studies, it has been recently discovered that the aerosols of patients being infected with Sars-CoV-2, might contain viable virus concentrations which are large enough to cause further infections if somebody else inhales those aerosols [5–7]. With half-life periods of the virus on aerosols on the order of 1 hour [8], it becomes evident that proper ventilation, strongly reducing the aerosol concentration, can help to prevent hidden infections, i.e. infections where the infected person is not (yet) aware of their infection but already contagious. Since aerosols and  $CO_2$  are both parts of exhaled air, measuring the  $CO_2$  concentration in a room provides an easy accessible indicator for the aerosol concentration [9].

Here we present a simple and cost effective, yet reliable way to monitor the CO<sub>2</sub> concentration. Widely available microelectronic components are used which can be easily programmed via open source software platforms allowing to modify and extend the example presented in this paper. Students can build the detectors in class as a joint project which might serve to raise interest in electronics or the underlying physical and chemical processes [10].

This work was inspired by a project of the *Hochschule Trier* [11], where the design and construction of a CO<sub>2</sub> measuring device is suggested as a students' project, allowing to discuss a variety of scientific topics during the course of the project. In addition, a few posts from different forums served as an inspiration [12–15]. Furthermore, a small number of GitHub repositories using the same CO<sub>2</sub> detector are available [16–18] (we would like to recommend the interested reader in particular to the repository by paulvha [18] as it contains a rather large number of examples).

# 2 The $CO_2$ monitor

The CO<sub>2</sub> monitor is based on the microelectronic sensor SCD30 which measures the CO<sub>2</sub> concentration and also provides measurements of the ambient temperature and relative humidity [19]. Using Arduino as a programing language and some microcontroller, it is straightforward to get the sensor running and outputting data, thanks to the examples available in the libraries provided by SparkFun [20]. Using the Arduino IDE [21], which is available for all major operating systems, the corresponding libraries can be simply included via the library manager.

To make the CO<sub>2</sub> monitor visually appealing, we decided to output the measurement to an OLED display (which is very inexpensive and available

in a large variety of sizes and configurations). Due to the widespread usage of such displays, they can also be directly included via the library manager in the Arduino IDE. In addition to just showing some numbers, we have included a red LED which lights up as soon as some threshold value of the CO<sub>2</sub> concentration is reached, indicating the need for ventilation. One could also think of a traffic light design, where first a yellow LED lights up at a slightly lower threshold value. The Federation of European Heating, Ventilation and Air Conditioning associations (REHVA) recommend to issue a warning, corresponding to an orange light, when a value of 800 ppm is reached and prompt to trigger some action like ventilation, corresponding to a red light, when 1000 ppm are reached [22]. The Federal Ministry of Labour and Social Affairs of Germany also states a threshold value of 1000 ppm that should not be passed [23]. Note that a value of approximately 410 ppm is the typical CO<sub>2</sub> concentration of air [24].

As controller we decided to use the low-cost open source NodeMCU board [25], as it offers enough flexibility to further extend the functionality of the  $CO_2$  monitor. Of particular interest might be the WiFi capability allowing for example to write the measured values to a web-server where they can then accessed via a web-browser or an app on a smartphone.


A prototype of the  $CO_2$  monitor is shown in Fig. 1. As one can see, it is not enclosed in some box to still allow easy access for modifications. The idea of this prototype was rather to show that the general principle of the  $CO_2$  monitor is working and not to provide a polished final product. The prototype is ready to be used in a class room or lecture hall, although it might be worth to mount everything into a box which is not only visually more appealing but provides also some protection.

## 3 Required parts

The CO<sub>2</sub> monitor as presented here consists of a number of parts for which it is not important to use the exact same model. The only component which should not be replaced is the CO<sub>2</sub> measuring device, the SCD30. Note that the program code discussed in Sec. 6 is tailored for the NodeMCU ESP8266, replacing that component would thus require small adjustments to the code.

The parts used for the prototype of the  $CO_2$  monitor are listed in Table 3. The display can be easily replaced by an OLED of larger size. One could also use multiple displays, which would require to take care of proper addressing the displays and thus add a little bit of complexity to the code (and to the assembly).

The usage of a breadboard was motivated by educational purposes as






Figure 1: Assembled and working prototype of the  $CO_2$  monitor, (*left*) with a measured  $CO_2$  concentration below the threshold and (*right*) above it (note the red LED).

this allows very easy assembly without the need to solder anything. It can, however, directly be replaced by a stripboard or completely omitted and use only cables or pin headers (which would require some soldering).

Note that the prices as listed in the table can be pushed down (significantly for some of the components) when ordering larger quantities.

For the prototype design of the CO<sub>2</sub> monitor we have decided to leave out a proper casing. One could either use a standard-sized case, or design one and print it for example on a 3D printer or re-use/recycle some old boxes. It is however important to correctly position the SCD30 inside the box: as described in a manufacturer's document [26], the sensor is ideally placed as close as possible to the box's outer shell and to a large opening to be properly exposed to the ambient. The box should be as small as possible to get fast response times to changes in the ambient air. The SCD30 should also be isolated from direct air flow, as the corresponding changes in pressure (due to the air flow) would lead to increased noise and thus reduced accuracy in the measurements. It is also recommended to not directly place the sensor above heat sources like for example microcontrollers.

# 4 The $CO_2$ sensor

The SCD30 has been chosen because it performs direct measurements of the  $CO_2$  concentration. Cheaper sensors often measure the concentration

| Element               | Quantity | Price |
|-----------------------|----------|-------|
| $SCD30 (CO_2 sensor)$ | 1        | 45€   |
| NodeMCU EPS8266       | 1        | 8€    |
| 0.91" OLED display    | 1        | 5€    |
| red LED               | 1        | 0.2€  |
| 220 Omega resistor    | 1        | .1€   |
| mini breadboard       | 1        | 4€    |
| breadboard cables     | 10       | 4€    |
| pin header            | 1        | 0.5€  |
| micro USB cable       | 1        | 3€    |

Table 1: Components used for the  $CO_2$  monitor as presented in this paper (note that the prices were obtained in 09/2020 and may vary).

of volatile organic compounds (VOC) and then assume a correlation between the two quantities. This can, however, lead to wrong values of the CO<sub>2</sub> concentration since VOC can be emitted from a variety of chemicals. Although VOCs are also known to cause health problems, here we are explicitly interested in the CO<sub>2</sub> concentration, as discussed in Sec. 1. For a discussion about monitoring VOC and CO<sub>2</sub> concentration with self-assembled devices we would like to point the interested reader to e.g. Ref. [27].

#### 4.1 Technical specifications

According to the datasheet of the SCD30 [19], the  $CO_2$  sensor has a measurement range of  $0-40,000\,\mathrm{ppm}$  with an accuracy of  $\pm 30\,\mathrm{ppm}$ . The supply voltage needs to be between 3.3 and 5 V which allows to use a variety of microcontrollers. The drawn current is specified to be on average 19 mA with a maximum value of 75 mA. With a sensor lifetime of 15 years, the SCD30 offers a reliable system to permanently monitor the  $CO_2$  concentration.

## 4.2 Nondispersive infrared technique

The CO<sub>2</sub> concentration is measured using the so-called nondispersive infrared technique (NDIR). It is the most common sensor type used in industry to measure the CO<sub>2</sub> concentration. Its principle is sketched in Fig. 2. A light source emits infrared light which travels through a tube filled with a sample of the surrounding air. The spectrum of the emitted light includes the  $4.26 \,\mu\mathrm{m}$  absorption band of CO<sub>2</sub> which is unique to the typical components of air and the light is absorbed by them. At the end of the tube, the remaining light




Figure 2: Sketch of a sensor using the nondispersive infrared technique to measure  $CO_2$  concentration.

hits an optical filter that allows only that specific wavelength of  $4.26\,\mu\mathrm{m}$  to pass. A detector then collects the remaining light. The difference between the amount of light emitted by the source and received by the detector is due to the  $\mathrm{CO}_2$  molecules in the tube which then allows to calculate the  $\mathrm{CO}_2$  concentration.

Using folded optics, i.e. waveguides, for the tube and diodes for the infrared source and detector, allows for a very compact size of the overall sensor on the order of just a few centimeters.

# 5 Assembly




Figure 3: Schematic of a prototype of the  $CO_2$  monitor.

The CO<sub>2</sub> monitor can be assembled in various ways, here we will restrict

ourselves to the case of a simple prototype design on a breadboard as shown in Fig. 3. The connection between the NodeMCU (with the ESP8266) and the SCD30 sensor is as follows:

| NodeMCU         |                   | SCD30  |
|-----------------|-------------------|--------|
| GND             | $\longrightarrow$ | GND    |
| $3.3\mathrm{V}$ | $\longrightarrow$ | VIN    |
| D2/GPI04        | $\longrightarrow$ | RX/SDA |
| D1/GPI05        | $\longrightarrow$ | TX/SCL |
| GND             | $\longrightarrow$ | SEL    |

The NodeMCU then needs to be connected to the OLED display as follows:

| NodeMCU         |                   | OLED display |
|-----------------|-------------------|--------------|
| GND             | $\longrightarrow$ | GND          |
| $3.3\mathrm{V}$ | $\longrightarrow$ | VCC          |
| D2/GPI04        | $\longrightarrow$ | SDA          |
| D1/GPIO5        | $\longrightarrow$ | SCL          |

It is of course also possible to directly connect the respective SDA and SCL pins of the OLED and the SCD30, as shown in Fig. 3, instead of connecting those pins between the SCD30 and the NodeMCU. The red LED is connected with its anode, the longer leg, to pin D8/GPI015 of the NodeMCU and with its cathode, the shorter leg, via a  $220\,\Omega$  resistor (to limit the current) to ground.

## 6 The program code

Arduino is used as programming language in this project due to its widespread usage and large numbers of libraries available for various hardware components. The Arduino IDE library manager allows to directly install a proper Arduino library for the SCD30. Alternatively, the library is available as a GitHub repository [20]. For a tutorial on how to install libraries within the Arduino IDE, see Ref. [28]. As for the NodeMCU and the OLED display, the Arduino IDE library manager is able to provide the required libraries.

The source code for the CO<sub>2</sub> monitor as described in this paper is available on GitHub [29], in order to be able to update and extend it. Nevertheless, we have also included the code in this paper, to provide a complete description

of the project. The include statements of the code are listed in Listing 1. The Adafruit\_GFX.h and Adafruit\_SSD1306.h libraries are used for the OLED display and are required to be installed via the library manager of the Arduino IDE beforehand (alternatively, they are also available on GitHub [30] for manual installation). Note that the display size in pixels needs to be set correctly and can vary. The SparkFun\_SCD30\_Arduino\_Library.h also needs to be installed via the library manager (or manually from the GitHub repository [20]).

```
1 #include <Wire.h>
                                       // for I2C communication
                                       // for writing to display
// for writing to display
 2 #include <Adafruit_GFX.h>
   #include <Adafruit_SSD1306.h>
  #include "SparkFun_SCD30_Arduino_Library.h"
6 // set to true if WiFi is desired, otherwise corresponding code is not
       compiled
  #define WIFI_ENABLED true
9 #if WIFI_ENABLED
10
    #include <ESP8266WiFi.h>
                                       // for SHA1 algorith (for Font Awesome)
    #include <Hash.h>
11
12
    #include <ESPAsyncTCP.h>
    #include <ESPAsyncWebServer.h>
13
    #include "Webpageindex.h"
14
                                       // webpage content, same folder as .ino
15
    // Replace with your network credentials
    const char* ssid = "ENTER_SSID";
const char* password = "ENTER_PASSWORD";
17
18
19 #endif
20
21 // activate debugging
22 // true: print info + data to serial monitor
23 // false: serial monitor is not used
24 #define DEBUG true
25
                                       // threshold for warning
26 #define CO2_CRITICAL 1500
   #define WARNING_DIODE_PIN D8
                                       // NodeMCU pin for red LED
27
29 #define MEASURE_INTERVAL 10
                                       // seconds, minimum: 2
30
31 #define SCREEN_WIDTH 128
                                       // OLED display width in pixels
32 #define SCREEN_HEIGHT 32
                                       // OLED display height in pixels
33
34 // OLDE reset pin, 4 is default (-1 if sharing Arduino reset pin)
35 // using NodeMCU, we have to use LED_BUILTIN
36 #define OLED_RESET LED_BUILTIN
_{
m 37} // Declaration for an SSD1306 display connected to I2C (SDA, SCL pins)
38 Adafruit_SSD1306 display(SCREEN_WIDTH, SCREEN_HEIGHT, &Wire, OLED_RESET);
39
40 SCD30 airSensor;
41 // use "unsigned long" for variables that hold time
42 // --> value will quickly become too large for an int
43 unsigned long previousMilliseconds = 0; // store last time scd30 was
       updated
44
45 // update scd30 readings every MEASURE_INTERVAL seconds
46 const long interval = MEASURE_INTERVAL*1000;
```

Listing 1: Loading the required libraries.

A switch is included in the header of the code allowing to enable or disable WiFi capabilities (by setting the variable WiFi\_ENABLED respectively to true or false). The libraries required for using WiFi are only included if the corresponding switch is set to true. In this example, we decided to use the ESPAsyncWebServer [31], based on ESPAsyncTCP [32], for a webserver supposed to run on the ESP8266 because asynchronous networks, as provided by these two libraries, allow us to handle more than just one connection at a time (which is important if used in a classroom environment). During the time of writing this article, these libraries require manual installation, i.e. getting a zip file from the GitHub repositories and include those zip files manually as libraries in the Arduino IDE.

To display the values measured by the SCD30 sensor on a website, we use global variables in the code, as shown in Listing 2. The complete html code for the website is loaded via including it in as a library and then copying into a string variable, called webpage.

```
#if WIFI_ENABLED
    // temperature, humidity, CO2 for web-page, updated in loop()
    float temperature_web = 0.0;
    float humidity_web
                         = 0.0;
                           = 0.0;
    float co2_web
    // create AsyncWebServer object on port 80
    AsyncWebServer server(80);
    // read html into string
    String webpage = index_html;
    // function for replacing placeholder on webpage with SCD30 values
13
    String processor(const String& var){
14
      //Serial.println(var);
      if (var == "CO2"){
        return String(co2_web);
18
      else if(var == "TEMPERATURE"){
19
20
        return String(temperature_web);
21
      else if(var == "HUMIDITY"){
22
23
        return String(humidity_web);
24
25
      return String();
    }
26
  #endif
```

Listing 2: Prepare website.

The code for the webpage is shown in Listing 3.

```
const char index_html[] PROGMEM = R"rawliteral(
2 <!DOCTYPE HTML><html>
```

```
3 <head>
    <!-- make webpage fit to your browser, not matter what OS or browser -->
    <meta name="viewport" content="width=device-width, initial-scale=1">
    <!-- load Font Awesome, get integrity and url here: https://fontawesome.
      com/account/cdn -->
    <link rel="stylesheet" href="https://use.fontawesome.com/releases/v5.14.0/</pre>
      css/all.css" integrity="sha384-HzLeBuhoNPvS15KYnjxOBT+WBOQEEqLprO+
      NBkkk5gbc67FTaL7XIGa2w1L0Xbgc" crossorigin="anonymous">
    <!-- add some CSS style: font, size for header (h2) and paragraph (p) -->
    <1--
                             size and format for labels to read
    <style>
10
     html {
11
12
       font-family: Arial;
       display:
                    inline-block;
13
                    0px auto;
14
       margin:
15
       text-align: center;
16
17
      h2 {
18
        font-size: 3.0rem;
19
      p {
20
21
        font-size: 3.0rem;
22
      .units {
23
        font-size: 1.2rem;
24
25
      .scd30-labels{
26
        font-size:
                        1.5rem:
        vertical-align: middle;
28
        padding-bottom: 15px;
29
30
31
    </style>
32
  </head>
33
  <body>
34
    <h2>CO2 monitor</h2>
    <!-- paragraph for CO2 concentration -->
35
36
      <i class="fas fa-head-side-cough" style="color:#ff6600;"></i>
37
      <span class="scd30-labels">CO2 concentration</span>
38
      <span id="co2">%CO2%</span>
39
      <sup class="units">ppm</sup>
40
41
    <!-- paragraph for temperature -->
42
43
    >
      <i class="fas fa-thermometer-half" style="color:#059e8a;"></i></i>
44
      <span class="scd30-labels">Temperature </span>
45
      <span id="temperature">%TEMPERATURE%</span>
46
47
      <sup class="units">&#8451</sup>
    48
49
    <!-- paragraph for humidity -->
50
    >
      <i class="fas fa-tint" style="color:#00add6;"></i>
51
      <span class="scd30-labels">Humidity</span>
      <span id="humidity">%HUMIDITY%</span>
53
      <sup class="units">%</sup>
54
55
    <!-- paragraph for getting additional information (github) and getting
56
      into contact via twitter -->
57
      58
      <span style="font-size:1.0rem;">The CO2 monitor on </span>
59
    <a href="https://github.com/alfkoehn/CO2_monitor" target="_blank" style=</pre>
```

```
"font-size:1.0rem;">GitHub (documentation + code)</a>
     </P>
61
62
     >
       <i class="fab fa-twitter" style="font-size:1.0rem; color:#1DA1F2;"></i></i>
63
64
       <span style="font-size:1.0rem;">Twitter: </span>
       <a href="https://twitter.com/formbar" target="_blank" style="font-size</pre>
65
       :1.0rem;">@formbar</a>
     </P>
   </body>
67
68
   <!-- JavaScript to update CO2, temperature and humidity automatically -->
69 <script>
70 setInterval(function () {
71
     var xhttp = new XMLHttpRequest();
72
     xhttp.onreadystatechange = function() {
       if (this.readyState == 4 && this.status == 200) {
73
74
         document.getElementById("co2").innerHTML = this.responseText;
75
76
     };
     xhttp.open("GET", "/co2", true);
77
78
     xhttp.send();
79 }, 10000 );
80
   setInterval(function ( ) {
81
     var xhttp = new XMLHttpRequest();
     xhttp.onreadystatechange = function() {
83
       if (this.readyState == 4 && this.status == 200) {
84
         document.getElementById("temperature").innerHTML = this.responseText;
85
86
87
     xhttp.open("GET", "/temperature", true);
88
89
     xhttp.send();
   }, 10000 );
90
91
   setInterval(function ( ) {
92
93
     var xhttp = new XMLHttpRequest();
     xhttp.onreadystatechange = function() {
  if (this.readyState == 4 && this.status == 200) {
94
95
         document.getElementById("humidity").innerHTML = this.responseText;
96
97
98
     };
     xhttp.open("GET", "/humidity", true);
99
100
     xhttp.send();
101 }, 10000 )
102 </script>
103 </html>)rawliteral";
```

Listing 3: Code for the webpage.

Listing 4 shows the **setup** function of the code, where the serial monitor is initialized, followed by the diode, optionally the WiFi, the OLED display, and then the SCD30. Finally, the webserver and the functions required to update the data on the webpage are prepared.

```
void setup(){
   if (DEBUG == true) {
      // initialize serial monitor at baud rate of 115200
      Serial.begin(115200);
      Serial.println("Using SCD30 to get: CO2 concentration, temperature, humidity");
}
```

```
// initialize I2C
9
     Wire.begin();
10
     // initialize LED pin as an output
11
    pinMode(WARNING_DIODE_PIN, OUTPUT);
12
13
14 #if WIFI_ENABLED
    // Connect to Wi-Fi
16
    WiFi.begin(ssid, password);
     if (DEBUG == true)
17
      Serial.println("Connecting to WiFi");
18
19
     while (WiFi.status() != WL_CONNECTED) {
20
      delay(1000);
      if (DEBUG == true)
21
22
        Serial.print(".");
23
24
     if (DEBUG == true)
      Serial.println(WiFi.localIP());
25
  #endif
26
27
     // SSD1306_SWITCHCAPVCC: generate display voltage from 3.3V internally
28
    if(!display.begin(SSD1306_SWITCHCAPVCC, 0x3C)) { // Address 0x3C for 128
29
       if (DEBUG == true)
30
         Serial.println(F("SSD1306 allocation failed"));
31
                                      // don't proceed, loop forever
32
       for(;;);
33
34
                                       // initialize display
    display.display();
35
36
                                       // library will show Adafruit logo
37
     delay(2000);
                                       // pause for 2 seconds
                                       // clear the buffer
38
    display.clearDisplay();
                                       //\  \, has\  \, to\  \, be\  \, set\  \, initially
39
     display.setTextSize(1);
40
     display.setTextColor(WHITE);
                                       // has to be set initially
41
42
     // move cursor to position and print text there
43
    display.setCursor(2,5);
    display.println("CO2 monitor");
44
    display.println("twitter.com/formbar");
45
  #if WIFI_ENABLED
46
47
    display.println(WiFi.localIP());
48
    display.println("WiFi disabled");
49
  #endif
50
51
    display.display();
                                       // write display buffer to display
53
     // turn warning LED on and off to test it
54
    digitalWrite(WARNING_DIODE_PIN, HIGH);
     delay(2000);
56
    digitalWrite(WARNING_DIODE_PIN, LOW);
57
58
59
     // initialize SCD30
     // SCD30 has data ready every two seconds
60
61
     if (airSensor.begin() == false) {
62
      if (DEBUG == true)
63
        Serial.println("Air sensor not detected. Please check wiring. Freezing
       ...");
      while (1)
64
65
66
```

```
68
  #if WIFI_ENABLED
69
     // Route for root / web page
     server.on("/", HTTP_GET, [](AsyncWebServerRequest *request){
  request->send_P(200, "text/html", index_html, processor);
70
71
72
     server.on("/co2", HTTP_GET, [](AsyncWebServerRequest *request){
73
       request->send_P(200, "text/plain", String(co2_web).c_str());
74
75
     });
76
     server.on("/temperature", HTTP_GET, [](AsyncWebServerRequest *request){
       request->send_P(200, "text/plain", String(temperature_web).c_str());
77
78
     server.on("/humidity", HTTP_GET, [](AsyncWebServerRequest *request){
79
80
       request->send_P(200, "text/plain", String(humidity_web).c_str());
81
     });
82
     // Start server
     server.begin();
83
84 #endif
85
```

Listing 4: Setup code.

The main code, the loop function, is given in Listing 5. First, the data is obtained from the SCD30 sensor and then passed to a function outputting it to the serial monitor and then to another function, printing it on the OLED display. Listings 6 and 7 show the code for the two latter functions. The data is then copied into the corresponding global variables to prepare the next update for the webpage. Finally, it is checked if the CO<sub>2</sub> concentration is above a critical threshold: a red LED indicates too high a value in our example (one could also think of an acoustic signal and some visual change on the webpage).

```
void loop(){
    float
      co2 new.
      temperature_new,
      humidity_new;
    unsigned long currentMilliseconds;
    // get milliseconds passed since program started to run
10
11
    currentMilliseconds = millis();
12
    if (currentMilliseconds - previousMilliseconds >= interval) {
13
14
      // save the last time you updated the DHT values
      previousMilliseconds = currentMilliseconds;
15
16
      if (airSensor.dataAvailable()) {
17
        // get data from SCD30 sensor
18
        co2_new
                     = airSensor.getCO2();
19
        temperature_new = airSensor.getTemperature();
        humidity_new
                       = airSensor.getHumidity();
21
22
23
         // print data to serial console
        if (DEBUG == true)
24
          printToSerial(co2_new, temperature_new, humidity_new);
```

```
27
         // print data to OLED display
28
         printToOLED(co2_new, temperature_new, humidity_new);
29
  #if WIFI_ENABLED
30
         // updated values for webpage
31
        co2_web
32
                        = co2_new;
         temperature_web = temperature_new;
33
        humidity_web
                       = humidity_new;
34
35
  #endif
36
37
38
       // if CO2-value is too high, issue a warning
      if (co2_new >= CO2_CRITICAL) {
39
         digitalWrite(WARNING_DIODE_PIN, HIGH);
40
41
         digitalWrite(WARNING_DIODE_PIN, LOW);
42
43
    }
44
45 }
```

Listing 5: Main loop which is executed repeatedly.

```
void printToSerial( float co2, float temperature, float humidity) {
   Serial.print("co2(ppm):");
   Serial.print(co2, 1);
   Serial.print(" temp(C):");
   Serial.print(temperature, 1);
   Serial.print(" humidity(%):");
   Serial.print(humidity, 1);
   Serial.print(humidity, 1);
}
```

Listing 6: Function which prints data to the serial console.

```
void printToOLED( float co2, float temperature, float humidity) {
    {\tt int}
      x0, x1;
                         // to align output on OLED display vertically
    x0 = 2;
    x1 = 86;
    display.clearDisplay();
    display.setCursor(x0,5);
9
     display.print("CO2 (ppm):");
10
     display.setCursor(x1,5);
11
     // for floats, 2nd parameter in display.print sets number of decimals
12
13
     display.print(co2, 0);
14
     display.setCursor(x0,15);
15
     display.print("temp. ( C)");
16
17
     display.setCursor(x0+7*6,15);
18
     display.write(167);
     display.setCursor(x1,15);
19
    display.print(temperature, 1);
20
21
    display.setCursor(x0,25);
22
    display.print("humidity (%):");
23
24
     display.setCursor(x1,25);
    display.print(humidity, 1);
```

```
26 display.display();
28 }
```

Listing 7: Function which prints data to the OLED display.

# Acknowledgments

The author is indebted to the efforts of the open-source software community.

#### References

- D. Twardella, W. Matzen, T. Lahrz, R. Burghardt, H. Spegel, L. Hendrowarsito, A. C. Frenzel, and H. Fromme. Effect of classroom air quality on students' concentration: results of a cluster-randomized crossover experimental study. *Indoor Air*, 22:378, 2012, doi:10.1111/j.1600-0668.2012.00774.x.
- [2] S. Gaihre, S. Semple, J. Miller, S. Fielding, and S. Turner. Classroom Carbon Dioxide Concentration, School Attendance, and Educational Attainment. J. Sch. Health, 85:569–574, 2014, doi:10.1111/josh.12183.
- [3] D. G. Shendell, R. Prill, W. J. Fisk, M. G. Apte, D. Blake, and D. Faulkner. Associations between classroom co<sub>2</sub> concentrations and student attendance in washington and idaho. *Indoor Air*, 14:333, 2004, doi:10.1111/j.1600-0668.2004.00251.x.
- [4] J. G. Allen, P. MacNaughton, U. Satish, S. Santanam, J. Vallarino, and J. D. Spengler. Associations of Cognitive Function Scores with Carbon Dioxide, Ventilation, and Volatile Organic Compound Exposures in Office Workers: A Controlled Exposure Study of Green and Conventional Office Environments. *Environmental Health Perspectives*, 6:805, 2016, doi:10.1289/ehp.1510037.
- [5] J. A. Lednicky, M. Lauzardo, Z. Hugh Fan, A. S. Jutla, T. B. Tilly, M. Gangwar, M. Usmani, S. N. Shankar, K. Mohamed, A. Eiguren-Fernandez, C. J. Stephenson, Alam. Md. M., M. A. Elbadry, J. C. Loeb, K. Subramaniam, T. B. Waltzek, K. Cherabuddi, J. G. Morris Jr., and C.-Y. Wu. Viable sars-cov-2 in the air of a hospital room with covid-19 patients. medRxiv, 2020, doi:10.1101/2020.08.03.20167395.

- [6] L. Morawska and D. K. Milton. It is time to address airborne transmission of covid-19. Clinical Infectious Diseases, 2020, doi:10.1093/cid/ciaa939.
- [7] M. A. Kohanski, L. J. Lo, and M. S. Waring. Review of indoor aerosol generation, transport, and control in the context of covid19. *International Forum of Allergy & Rhinology*, 2020, doi:10.1002/alr.22661.
- [8] N. v. Doremalen, T. Bushmaker, D. H. Morris, M. G. Holbrook, A. Gamble, B. N. Williamson, A. Tamin, J. L. Harcourt, N. J. Thornburg, S. I. Gerber, J. O. Lloyd-Smith, E. de Witt, and V. J. Munster. Aerosol and Surface Stability of SARS-CoV-2 as Compared with SARS-CoV-1. N. Engl. J. Med., 382:1564–1567, 2020, doi:10.1056/NEJMc2004973.
- [9] A. Hartmann and M. Kriegel. Risk assessment of aerosols loaded with virus based on co<sub>2</sub>-concentration. *Preprint server of the Technische Universität Berlin*, 2020, doi:10.14279/depositonce-10362.
- [10] B. Watzka and R. Girwidz. Ist die Luft zu schlecht zum Lernen? Nichtdispersive IR-CO2-Gassensoren im Physikunterricht. *Physik und Didak*tik in Schule und Hochschule, 10:22–33, 2011.
- [11] Hochschule Trier Umwelt-Campus Birkenfeld. Website: Covid-19 Prävention: CO2-Messung und bedarfsorientierte Lüftung (accessed: 2020-09-13). https://www.umwelt-campus.de/forschung/projekte/iot-werkstatt/ideen-zur-corona-krise.
- [12] hackster.io (user Brian Boyles). Website: Homemade CO2 Sensor Unit (accessed: 2020-09-13). https://www.hackster.io/bfboyles/homemade-co2-sensor-unit-22a9d8.
- [13] Arduino Forum (user metropol). Website: Arduino forum: Co2 level classroom sensor (accessed: 2020-09-13). https://forum.arduino.cc/index.php?topic=702131.
- [14] Home Assistant Forum (user Ombra). Website: Co2 monitoring with scd30 sensor and ha integration (accessed: 2020-09-13). https://community.home-assistant.io/t/co2-monitoring-with-scd30-sensor-and-ha-integration/216708.
- Ρi (user Website: En-[15] Raspberry Forum Zentris). In twicklung: Co2-Sensor (SCD30)mit ESP8266 und Display SSD1306 (kurze Vorstellung) (accessed: 2020-09-13). https://forum-raspberrypi.de/forum/thread/ 44717-in-entwicklung-co2-sensor-scd30-mit-esp8266-und-display-ssd1306-kurze

- [16] Kaspar Metz. Coro<sub>2</sub>sens. https://github.com/kmetz/coro2sens, 2020.
- [17] Netzbasteln. Co<sub>2</sub>narienvogel. https://github.com/netzbasteln/co2narienvogel, 2020.
- [18] paulvha. paulvha scd30 library. https://github.com/paulvha/scd30, 2020.
- [19] Sensirion: The Sensor Company. Datasheet sensirion scd30 sensor module (version 0.94). https://www.sensirion.com/fileadmin/user\_upload/customers/sensirion/Dokumente/9.5\_CO2/Sensirion\_CO2\_Sensors\_SCD30\_Datasheet.pdf.
- [20] SparkFun Electronics. SparkFun SCD30 CO2 Sensor Library. https://github.com/sparkfun/SparkFun\_SCD30\_Arduino\_Library, 2020.
- [21] The Arduino team. Software: The open-source arduino ide. https://www.arduino.cc/en/main/software.
- [22] Ventilation The Federation of European Heating and Air Conditioning associations (REHVA). Website: Rehva covid-19 guidance document (version 2020-08-03). hhttps://www.rehva.eu/activities/covid-19-guidance/rehva-covid-19-guidance.
- [23] Federal Ministry of Labour and Social Affairs Germany. Website: SARS-CoV-2-Arbeitsschutzregel (version 2020-08-20). https://www.bmas.de/DE/Presse/Meldungen/2020/neue-sars-cov-2-arbeitsschutzregel.html.
- [24] National Oceanic and Atmospheric Administration (NOAA). Website: Earth system research laboratory (accessed: 2020-09-28). https://www.esrl.noaa.gov/gmd/ccgg/trends/.
- [25] NodeMCU. Website: Nodemcu documentation (accessed: 2020-09-18). https://nodemcu.readthedocs.io/en/master/.
- [26] Sensirion: The Sensor Company. Design-in guidelines for scd30 (version 0.2). https://www.sensirion.com/fileadmin/user\_upload/customers/sensirion/Dokumente/9.5\_CO2/Sensirion\_CO2\_Sensors\_SCD30\_Assembly\_Guideline.pdf.
- [27] G. Chiesa, S. Cesari, M. Garcia, M. Issa, and S. Li. Multisensor IoT Platform for Optimising IAQ Levels in Buildings through a Smart Ventilation System. Sustainability, 1:5777, 2019, doi:10.3390/su11205777.

- [28] SparkFun Electronic. Website: Installing an Arduino Library. https://learn.sparkfun.com/tutorials/installing-an-arduino-library.
- [29] Alf Köhn-Seemann. Co<sub>2</sub> monitor. https://github.com/alfkoehn/CO2\_monitor, 2020.
- [30] Adafruit Industries. Adafruit\_ssd1306. https://github.com/adafruit\_SSD1306, 2020.
- [31] me-no dev. Espasyncwebserver. https://github.com/me-no-dev/ESPAsyncWebServer, 2020.
- [32] me-no dev. Espasynctcp: Async tcp library for esp8266 arduino. https://github.com/me-no-dev/ESPAsyncTCP, 2020.